

FUNCTIONAL PEARL

Compiling a Fifty Year Journey

GRAHAM HUTTON

School of Computer Science, University of Nottingham, UK
(*email: gmh@cs.nott.ac.uk*)

PATRICK BAHR

Department of Computer Science, IT University of Copenhagen, Denmark
(*email: paba@itu.dk*)

Abstract

Fifty years ago, John McCarthy and James Painter (1967) published the first paper on compiler verification, in which they showed how to formally prove the correctness of a compiler that translates arithmetic expressions into code for a register-based machine. In this article, we revisit this example in a modern context, and show how such a compiler can now be calculated directly from a specification of its correctness using simple equational reasoning techniques.

1 Introduction

The first compiler correctness proof was published fifty years ago in a seminal paper by McCarthy and Painter (1967). The input to their compiler was arithmetic expressions built up from integers, variables and addition, and the output was code for a virtual machine with an accumulator and an infinite number of additional registers.

Correctness of the compiler was then specified in the following manner. If variables are stored in registers numbered below some value r , so that registers from r and above can be used for temporary storage, then executing the compiled code for an expression returns the value of the expression in the accumulator. Moreover, no registers except the accumulator and those from r and above are affected. Correctness was proved by induction on the expression being compiled, using a large number of lemmas.

In this article, we revisit this example in a modern context, and show how a compiler for arithmetic expressions can now be calculated directly from a specification of its correctness, using simple equational reasoning techniques and a few basic assumptions about the behaviour of registers. In fact we go further, showing how both the compiler and the virtual machine can be calculated from the same specification.

The methodology we use is based upon our recent work on calculating compilers for stack machines (Bahr & Hutton, 2015), adapted here to register machines. Our approach is founded on three ideas beyond those that were used by McCarthy and Painter: defining functions by pattern matching, which avoids the need for explicit destructors; generalising

the compiler to take additional code to be executed, which avoids the need for distributivity lemmas; and having the virtual machine ‘free up’ any temporary registers that were used, which avoids the need for reasoning modulo unused registers.

All our programs and calculations are written in Haskell, but we only use the basic concepts of recursive types, recursive functions, and inductive proofs. All of our calculations have been mechanically checked using the Coq proof assistant, and the proof scripts are available as online supplementary material for the article.

2 Preliminaries

The source language for our compiler is a datatype *Expr* of arithmetic expressions built up from integer values using an addition operator, whose semantics is given by a recursive function *eval* that evaluates an expression to its integer value:

```
data Expr = Val Int | Add Expr Expr
eval :: Expr → Int
eval (Val n)    = n
eval (Add x y) = eval x + eval y
```

For simplicity the source language does not include variables, but they pose no difficulties for our methodology. In turn, the target for our compiler is a virtual machine with a memory that comprises an infinite sequence of registers. Each register has a unique name and is either empty or stores a single integer value. To simplify our calculations, we use abstract types *Mem* and *Reg* for memories and register names:

```
type Mem = ...
type Reg = ...
```

We further assume that the machine has a special register called the accumulator, which is used to store the result of evaluating an expression. Rather than using a specific register in the memory for this purpose, we factor out the accumulator as a component of the *State* of the machine, which comprises the current accumulator value and memory:

```
type State = (Int, Mem)
```

Given the assumptions above, our goal now is to calculate three additional components that together complete the definition of a compiler for expressions:

- A datatype *Code* that represents code for the virtual machine;
- A function *compile* :: *Expr* → *Code* that compiles expressions to code;
- A function *exec* :: *Code* → *State* → *State* that provides a semantics for code.

Intuitively, *Code* will provide a suitable collection of primitive machine instructions for evaluating expressions, *compile* will translate an expression into a sequence of such instructions, and *exec* will execute code starting from an initial state of the machine to give a final state. Moreover, the desired relationship between the source language, compiler and virtual machine is captured by the following correctness property:

$$\text{exec } (\text{compile } e) (a, \text{empty}) = (\text{eval } e, \text{empty}) \quad (1)$$

That is, compiling an expression and executing the resulting code starting with an empty memory gives a final state in which the accumulator contains the value of the expression and the memory remains empty. In practice, the fact that the memory must also be empty in the final state means that if the compiled code for an expression uses any registers to store intermediate results, they must be freed up (returned to empty) afterwards, to ensure that the memory is returned to the original configuration. Once again, the rationale for this decision is that it simplifies our subsequent calculations.

Equation (1) captures the correctness of the compiler, but on its own isn't suitable as a basis for calculating the three undefined components. In particular, our methodology is based on induction, and as is often the case, we first need to generalise the property we are considering. We begin by generalising (1) from the *empty* memory to an arbitrary memory m , as the use of a specific memory would preclude the use of induction:

$$\text{exec}(\text{compile } e)(a, m) = (\text{eval } e, m)$$

Secondly, in order that the compiler can use registers for temporary storage, we assume the memory m is free from a given register r onwards, written as *isFreeFrom* r m , and generalise to a compilation function $\text{comp} :: \text{Expr} \rightarrow \text{Reg} \rightarrow \text{Code}$ that takes the first free register r as an additional argument, resulting in the following specification:

$$\text{isFreeFrom } r \ m \Rightarrow \text{exec}(\text{comp } e \ r)(a, m) = (\text{eval } e, m)$$

Finally, as in our recent work on compiler calculation (Bahr & Hutton, 2015), we further generalise *comp* to take additional code to be executed after the compiled code. This step is a key aspect of our methodology and significantly streamlines the calculations. Using these ideas, the correctness of the generalised compilation function:

$$\text{comp} :: \text{Expr} \rightarrow \text{Reg} \rightarrow \text{Code} \rightarrow \text{Code}$$

can then be specified by the following implication:

$$\text{isFreeFrom } r \ m \Rightarrow \text{exec}(\text{comp } e \ r \ c)(a, m) = \text{exec } c(\text{eval } e, m) \quad (2)$$

That is, if the memory is assumed to be free from a given register onwards, then compiling an expression and executing the resulting code gives the same result as executing the additional code with the value of the expression in the accumulator. The behaviour of memory primitives such as *empty* and *isFreeFrom* is considered in the next section.

In summary, (1) and (2) provide specifications for the undefined components, and our goal is to calculate definitions that satisfy these properties. Given that the two specifications have four unknowns (*Code*, *compile*, *exec* and *comp*), this may seem like an impossible task. However, as we shall see, it can be achieved using simple equational reasoning.

3 Memory model

Before proceeding to the calculation itself, we formalise our assumptions about the abstract types *Mem* for memories and *Reg* for register names that we introduced. First of all, we assume the following primitive operations on these types:

$$\begin{aligned} \text{empty} &:: \text{Mem} \\ \text{set} &:: \text{Reg} \rightarrow \text{Int} \rightarrow \text{Mem} \rightarrow \text{Mem} \end{aligned}$$

$get :: Reg \rightarrow Mem \rightarrow Int$
 $first :: Reg$
 $next :: Reg \rightarrow Reg$
 $free :: Reg \rightarrow Mem \rightarrow Mem$

Intuitively, *empty* is the initial memory in which all registers are empty (contain no value), while *set* and *get* respectively change and fetch the integer value of a given register in the memory; we will only apply *get* to non-empty registers, with its behaviour for empty registers being left unspecified. In turn, *first* is the name of the first register, *next* gives the name of the next register, and *free* makes a register empty. Note that *isFreeFrom* is not included in the above list, as it is a meta-level predicate for reasoning purposes rather than being an operation on the memory of the virtual machine.

A simple way to realise the above memory model is to represent a register name as a natural number, the memory as a function from register names to their current value, and use a special value to represent a register that is empty. For the purposes of our calculations, however, we only require the following properties of the primitive operations, and don't need to be concerned with precisely how they are defined.

$$\begin{aligned}
 & isFreeFrom\ first\ empty && \text{(EMPTY MEMORY)} \\
 & get\ r\ (set\ r\ n\ m) = n && \text{(SET/GET)} \\
 isFreeFrom\ r\ m \Rightarrow & free\ r\ (set\ r\ n\ m) = m && \text{(SET/FREE)} \\
 isFreeFrom\ r\ m \Rightarrow & isFreeFrom\ (next\ r)\ (set\ r\ n\ m) && \text{(SET/ISFREEFROM)}
 \end{aligned}$$

These properties state in turn that: every register from the first onwards is free in the empty memory; setting a register and then getting its value gives the expected result; setting the first free register and then freeing it up returns the memory to its previous state; and finally, setting the first free register leaves all subsequent registers free. These properties are not complete, but suffice for our calculations and arose naturally during their development.

4 Compiler calculation

To calculate the compiler we proceed directly from specification (2) by structural induction on the expression argument e , using the desire to apply the induction hypotheses as the driving force for the calculation process. In each case, we aim to rewrite the left-hand side $exec\ (comp\ e\ r\ c)\ (a, m)$ of the equation into the form $exec\ c'\ (a, m)$ for some code c' , from which we can then conclude that the definition $comp\ e\ r\ c = c'$ satisfies the specification in this case. In order to do this, we will find that we need to introduce new constructors into the *Code* type, along with their interpretation by the function *exec*.

In the base case, $e = Val\ n$, we assume $isFreeFrom\ r\ m$, and calculate as follows:

$$\begin{aligned}
 & exec\ (comp\ (Val\ n)\ r\ c)\ (a, m) \\
 = & \{ \text{specification (2)} \} \\
 & exec\ c\ (eval\ (Val\ n), m) \\
 = & \{ \text{applying eval} \} \\
 & exec\ c\ (n, m)
 \end{aligned}$$

Now we appear to be stuck, as no further definitions can be applied. However, we are aiming to end up with a term of the form $exec\ c'\ (a,m)$ for some code c' . That is, in order to complete the calculation we need to solve the following equation:

$$exec\ c'\ (a,m) = exec\ c\ (n,m)$$

Note that we can't simply use this equation as a definition for $exec$, because n and c would be unbound in the body of the definition. The solution is to package these two variables up in the code argument c' (which can freely be instantiated as it is existentially quantified, whereas the other variables in the equation are universally quantified), by adding a new constructor to the *Code* datatype that takes these two variables as arguments,

$LOAD :: Int \rightarrow Code \rightarrow Code$

and defining a new equation for the function $exec$ as follows:

$$exec\ (LOAD\ n\ c)\ (a,m) = exec\ c\ (n,m)$$

That is, executing the code $LOAD\ n\ c$ proceeds by loading the integer n into the accumulator and then executing the code c , hence the choice of name for the new constructor. Using these ideas, it is now straightforward to complete the calculation:

$$\begin{aligned} & exec\ c\ (n,m) \\ = & \{ \text{definition of } exec \} \\ & exec\ (LOAD\ n\ c)\ (a,m) \end{aligned}$$

The final term now has the form $exec\ c'\ (a,m)$, where $c' = LOAD\ n\ c$, from which we conclude that the following definition satisfies specification (2) in the base case:

$$comp\ (Val\ n)\ r\ c = LOAD\ n\ c$$

That is, the code for an integer value simply loads the value into the accumulator and then continues with the additional code that is supplied. Note that for this case we didn't need to use the register argument r or the *isFreeFrom* assumption.

For the inductive case, $e = Add\ x\ y$, we assume *isFreeFrom* $r\ m$ and begin in the same way as above by first applying the specification and the evaluation function:

$$\begin{aligned} & exec\ (comp\ (Add\ x\ y)\ r\ c)\ (a,m) \\ = & \{ \text{specification (2)} \} \\ & exec\ c\ (eval\ (Add\ x\ y),m) \\ = & \{ \text{applying } eval \} \\ & exec\ c\ (eval\ x + eval\ y, m) \end{aligned}$$

At this point no further definitions can be applied. However, as we are performing an inductive calculation, we can make use of the induction hypotheses for the expressions x and y . To use the induction hypothesis for y , which is

$$isFreeFrom\ r'\ m' \Rightarrow exec\ (comp\ y\ r'\ c')\ (a',m') = exec\ c'\ (eval\ y, m')$$

we must satisfy the precondition for some register r' and memory m' , and rewrite the term being manipulated into the form $exec\ c'\ (eval\ y, m')$ for some code c' . That is, we need to

satisfy the precondition $isFreeFrom\ r'\ m'$ and solve the equation:

$$exec\ c'\ (eval\ y, m') = exec\ c\ (eval\ x + eval\ y, m)$$

We are free to instantiate r' , m' and c' in order to achieve these goals. First of all, we generalise from the specific values $eval\ x$ and $eval\ y$ in the equation to give:

$$exec\ c'\ (a, m') = exec\ c\ (b + a, m)$$

We can't use this equation as a definition for $exec$, because the variables c , b and m would be unbound in the body of the definition. However, we are free to instantiate c' and m' in order to solve the equation. We consider each unbound variable in turn:

- For c , the simplest option is to put it into the argument c' as we did in the base case, by adding a new constructor. If we attempted to put c into m' , this would require storing code in the memory, which is not supported by our memory model.
- For b , the simplest option is to put it into the memory m' , by assuming that it is stored in a register. If we attempted to put b into the code c' , this would require evaluating the argument expression x at compile-time to produce this value, whereas for a compiler we normally expect evaluation to take place at run-time.
- For m , the simplest option is also to put it into m' , by assuming that m can be derived from m' in some way. If we attempted to put m into c' , this would require storing the entire memory in the code, which is not what we expect from a compiler.

How should we satisfy the above requirements for m and b ? One might try simply equating the memories and assuming the value is stored in the first free register, that is take $m' = m$ and assume $get\ r\ m = b$. However, the latter assumption is not satisfiable as it conflicts with our top-level assumption $isFreeFrom\ r\ m$ that all registers from r onwards are free in m .

The simplest way to resolve this problem is to equate the two memories for all registers *except* register r , which in the case of m' should contain the value b to satisfy our requirement, and in the case of m should be free to satisfy our top-level assumption. The desired relationship between the two memories can then be captured by two assumptions, $get\ r\ m' = b$ and $free\ r\ m' = m$, using which the equation to be solved now has the form:

$$exec\ c'\ (a, m') = exec\ c\ (get\ r\ m' + a, free\ r\ m')$$

The variable r is now unbound on the right-hand side of the equation, but can readily be packaged up along with the variable c in the code argument c' by adding a new constructor to the *Code* datatype that takes these two variables as arguments,

$ADD::\ Reg \rightarrow Code \rightarrow Code$

and defining a new equation for the function $exec$:

$$exec\ (ADD\ r\ c)\ (a, m) = exec\ c\ (get\ r\ m + a, free\ r\ m)$$

That is, executing the code $ADD\ r\ c$ proceeds by adding the value of register r to the accumulator and then freeing up this register in the memory, hence the choice of name for the new constructor. Using our three local assumptions

- (a) $isFreeFrom\ r'\ m'$

(b) $get\ r\ m' = eval\ x$

(c) $free\ r\ m' = m$

we then continue the calculation as follows:

$$\begin{aligned}
 & exec\ c\ (eval\ x + eval\ y, m) \\
 = & \{ \text{assumptions (b) and (c)} \} \\
 & exec\ c\ (get\ r\ m' + eval\ y, free\ r\ m') \\
 = & \{ \text{definition of } exec \} \\
 & exec\ (ADD\ r\ c)\ (eval\ y, m') \\
 = & \{ \text{induction hypothesis for } y, \text{ assumption (a)} \} \\
 & exec\ (comp\ y\ r'\ (ADD\ r\ c))\ (a', m')
 \end{aligned}$$

We are now free to chose m' and r' to satisfy the assumptions (a), (b) and (c). The simplest approach is to define m' by setting register r in memory m to the value $eval\ x$, and define r' as the next free register after r . That is, we take $m' = set\ r\ (eval\ x)\ m$ and $r' = next\ r$. It is then easy to verify that these assignments discharge the assumptions:

(a):

$$\begin{aligned}
 & isFreeFrom\ r'\ m' \\
 \Leftrightarrow & \{ \text{applying } r' \text{ and } m' \} \\
 & isFreeFrom\ (next\ r)\ (set\ r\ (eval\ x)\ m) \\
 \Leftrightarrow & \{ \text{SET/ISFREEFROM property, } isFreeFrom\ r\ m \} \\
 & True
 \end{aligned}$$

(b):

$$\begin{aligned}
 & get\ r\ m' \\
 = & \{ \text{applying } m' \} \\
 & get\ r\ (set\ r\ (eval\ x)\ m) \\
 = & \{ \text{SET/GET property} \} \\
 & eval\ x
 \end{aligned}$$

(c):

$$\begin{aligned}
 & free\ r\ m' \\
 = & \{ \text{applying } m' \} \\
 & free\ r\ (set\ r\ (eval\ x)\ m) \\
 = & \{ \text{SET/FREE property, } isFreeFrom\ r\ m \} \\
 & m
 \end{aligned}$$

In summary, using the assignments for m' and r' that we have determined above, the term that we are manipulating now has the following form:

$$exec\ (comp\ y\ (next\ r)\ (ADD\ r\ c))\ (a', set\ r\ (eval\ x)\ m)$$

We are free to chose the new accumulator value a' at this point. With a view to now applying the induction hypothesis for x , which requires that the accumulator contains $eval\ x$, we simply take $a' = eval\ x$, resulting in the following term:

$$\text{exec } (\text{comp } y \text{ (next } r \text{) (ADD } r \text{ c)}) (\text{eval } x, \text{ set } r \text{ (eval } x \text{) } m)$$

We could now apply the induction hypothesis for x , because the memory satisfies the precondition for the register $\text{next } r$. However, doing so would yield the term

$$\text{exec } (\text{comp } x \text{ (next } r \text{) (comp } y \text{ (next } r \text{) (ADD } r \text{ c))}) (a, \text{ set } r \text{ (eval } x \text{) } m)$$

which cannot be rewritten into the desired form $\text{exec } c' (a, m)$, as there is no way of retrieving the value $\text{eval } x$ from the state of the machine. Therefore, we have to transform the state from $(\text{eval } x, \text{ set } r \text{ (eval } x \text{) } m)$ into $(\text{eval } x, m)$ before applying the induction hypothesis. That is, we need to solve the equation

$$\text{exec } c' (\text{eval } x, m) = \text{exec } (\text{comp } y \text{ (next } r \text{) (ADD } r \text{ c)}) (\text{eval } x, \text{ set } r \text{ (eval } x \text{) } m)$$

As in the case for y , we first generalise the equation, in this case by abstracting over the value $\text{eval } x$ and the code $\text{comp } y \text{ (next } r \text{) (ADD } r \text{ c)}$, to give the following:

$$\text{exec } c' (a, m) = \text{exec } c (a, \text{ set } r \text{ a } m)$$

We can't use this equation as a definition for exec , because the variables c and r would be unbound in the body of the definition. However, we are free to instantiate c' in order to solve the equation. As previously we can package r and c up in the code argument c' by adding a new constructor to the *Code* datatype,

$$\text{STORE} :: \text{Reg} \rightarrow \text{Code} \rightarrow \text{Code}$$

and defining a new equation for the function exec :

$$\text{exec } (\text{STORE } r \text{ c}) (a, m) = \text{exec } c (a, \text{ set } r \text{ a } m)$$

That is, executing the code $\text{STORE } r \text{ c}$ proceeds by storing the accumulator value in register r , hence the choice of name for the new constructor.

We then continue the calculation:

$$\begin{aligned} & \text{exec } (\text{comp } y \text{ (next } r \text{) (ADD } r \text{ c)}) (\text{eval } x, \text{ set } r \text{ (eval } x \text{) } m) \\ &= \{ \text{definition of } \text{exec} \} \\ & \text{exec } (\text{STORE } r \text{ (comp } y \text{ (next } r \text{) (ADD } r \text{ c))}) (\text{eval } x, m) \\ &= \{ \text{induction hypothesis for } x, \text{ assuming } \text{isFreeFrom } r' \text{ } m \} \\ & \text{exec } (\text{comp } x \text{ } r' \text{ (STORE } r \text{ (comp } y \text{ (next } r \text{) (ADD } r \text{ c))})}) (a', m) \end{aligned}$$

We are now free to choose the register r' and the new accumulator value a' to satisfy the inductive assumption $\text{isFreeFrom } r' \text{ } m$. The simplest approach is just to take $r' = r$ and $a' = a$, under which the inductive assumption reduces to our top-level assumption $\text{isFreeFrom } r \text{ } m$ and the machine state has the desired form (a, m) . The resulting term

$$\text{exec } (\text{comp } x \text{ } r \text{ (STORE } r \text{ (comp } y \text{ (next } r \text{) (ADD } r \text{ c))})}) (a, m)$$

now has the form $\text{exec } c' (a, m)$ for some code c' , from which we conclude that the following definition satisfies specification (2) in the inductive case:

$$\text{comp } (\text{Add } x \text{ } y) \text{ } r \text{ c} = \text{comp } x \text{ } r \text{ (STORE } r \text{ (comp } y \text{ (next } r \text{) (ADD } r \text{ c))})}$$

That is, the code for addition first computes the value of expression x and stores the resulting value in the first free register r , and then computes the value of expression y

and adds the resulting value to the contents of register r . Note that when compiling y the next free register becomes *next* r , because r itself is used to store the value of x .

Finally, we consider the top-level function $compile :: Expr \rightarrow Code$, whose correctness was specified by equation (1). In a similar manner to (2), we aim to rewrite the left-hand side $exec (compile e) (a, empty)$ into the form $exec c (a, empty)$ for some code c , from which we can then conclude that the definition $compile e = c$ satisfies the specification. In this case there is no need to use induction as simple calculation suffices, during which we introduce a new constructor $HALT :: Code$ to transform the term being manipulated into the required form so that specification (2) can then be applied.

$$\begin{aligned}
& exec (compile e) (a, empty) \\
= & \{ \text{specification (1)} \} \\
& (eval e, empty) \\
= & \{ \text{define: } exec \text{ HALT } (a, m) = (a, m) \} \\
& exec \text{ HALT } (eval e, empty) \\
= & \{ \text{specification (2), EMPTY MEMORY property} \} \\
& exec (comp e \text{ first HALT}) (a, empty)
\end{aligned}$$

In summary, we have calculated the following definitions:

```

data Code = LOAD Int Code | STORE Reg Code | ADD Reg Code | HALT
compile :: Expr → Code
compile e = comp e first HALT
comp :: Expr → Reg → Code → Code
comp (Val n) r c = LOAD n c
comp (Add x y) r c = comp x r (STORE r (comp y (next r) (ADD r c)))
exec :: Code → State → State
exec (LOAD n c) (a, m) = exec c (n, m)
exec (STORE r c) (a, m) = exec c (a, set r a m)
exec (ADD r c) (a, m) = exec c (get r m + a, free r m)
exec HALT (a, m) = (a, m)

```

This compiler is essentially the same as McCarthy and Painter's (1967) except that i) our compiler has been calculated directly from a high-level specification of its correctness, with all the above compilation machinery falling naturally out of the calculation process; and ii) their source language also includes variables, which haven't been considered here for simplicity, but don't pose any difficulties for our approach. The calculation for the language with variables is included as a Coq proof in the online supplementary material.

5 Reflection

The original compiler correctness proof of McCarthy and Painter (1967) is rather complex, using many lemmas. A methodology that can be used for calculating a compiler has to be simpler, otherwise one becomes lost in the technical details. Three main ideas form the foundation of the simplified methodology that made our calculation possible: using pattern matching rather than deconstructors, strengthening the induction hypothesis using additional code, and using a specification that requires freeing up unused registers.

The use of pattern matching improves the clarity of the reasoning by virtue of providing a more compact notation than deconstructors. But pattern matching is also indispensable for the methodology for calculating the virtual machine. As part of the calculation we need to solve equations in order to make progress, and solving such an equation directly yields a case for the definition of the virtual machine by pattern matching.

The desire to apply induction hypotheses is the main driving force in our calculation process. Strengthening the induction hypothesis by introducing an additional code argument to the compiler means that the induction hypothesis becomes directly applicable without the need for additional lemmas (Hutton, 2016, chapter 16). The need for such lemmas would obscure the goal to which the calculation is targeted. Instead, the more general induction hypothesis allows us to make progress in the calculation by solving simple equations.

McCarthy and Painter use reasoning modulo unused registers. Rather than using normal equality, they reason with an equivalence relation $=_r$ on the memory of their machine, under which $m =_r m'$ when m and m' coincide on all registers prior to r . This relaxed equality allows the compiler to use registers from r onwards to store intermediate results. However, this approach makes reasoning with additional code more difficult. In particular, the specification for *comp* then becomes $exec\ (comp\ e\ r\ c)\ (a, m) = exec\ c\ (eval\ e, m')$, where the final memory m' is existentially quantified with the side condition that $m =_r m'$. Calculations with existentially quantified variables that are subject to side conditions are difficult to manage and prone to errors. Instead, our approach is to demand the equality of m and m' . As a consequence, the virtual machine has to ‘clean up after itself’ by using *free* to restore memory that has been used to store intermediate results.

One might think that this requirement may result in a less efficient implementation. However, all uses of *free* can safely be removed from the virtual machine. Or equivalently, we can instantiate the memory model with an implementation where *free r* is the identity function. We only use *free* to impose structure on the memory to simplify the reasoning. This structure, in the form of the *isFreeFrom* predicate, is not used in the virtual machine itself. Importantly, however, having a virtual machine that cleans up after itself allowed us to adapt the compiler calculation methodology that we developed for stack-based machines (Bahr & Hutton, 2015), which relies on the intrinsic structure of the stack.

In future work we plan to explore how our methodology can be scaled up to more expressive source and target languages, and how mechanical tool support can be used to assist with the resulting calculations and certify their correctness.

Acknowledgements We would like to thank Ralf Hinze, Jeremy Gibbons, and the JFP referees for many useful comments and suggestions. Graham Hutton was funded by EPSRC grant EP/P00587X/1, *Unified Reasoning About Program Correctness and Efficiency*.

References

- Bahr, Patrick, & Hutton, Graham. (2015). Calculating Correct Compilers. *Journal of Functional Programming*, 25(Sept.).
- Hutton, Graham. (2016). *Programming in Haskell*. Cambridge University Press.
- McCarthy, John, & Painter, James. (1967). Correctness of a Compiler for Arithmetic Expressions. *Pages 33–41 of: Mathematical Aspects of Computer Science*. Proceedings of Symposia in Applied Mathematics, vol. 19. American Mathematical Society.